Problem 15.7

Young and Freedman 14th Edition.

Given:

Transverse waves on a string have a wave speed of 8.00 m/s, amplitude 0.0700 m, and wavelength 0.320 m. The waves travel in the -x direction, and at t = 0, the x=0 end of the string has its maximum upward displacement.

```
In[40]:= v = 8.00; A = 0.0700; λ = 0.320;
```

(a) Compute related quantities f, T, k

```
In[41]:= T = λ / v
Out[41]= 0.04
```

```
In[42]:= f = 1/T
Out[42]= 25.
```

```
In[43]:= ω = 2π / T
Out[43]= 157.08
```

```
In[44]:= k = 2π / λ
Out[44]= 19.635
```

(b) Wave function

The equation for the traveling wave is the following. The ‘+‘ sign is because the wave travels to the left. The Cos[ ] is because we are given the initial condition y[0, 0] = A.

```
In[45]:= y[x_, t_] := A Cos[k x + ω t]
```

Here is a snapshot of the wave at time t = 0.

In[46]:=
```
Plot[y[x, 0], {x, 0, 3 λ},
 AxesLabel → {"t (s)", "Amp (m)"}, LabelStyle → Larger]
```

Out[46]=



In[47]:=
```
Animate[Plot[y[x, t], {x, 0, 3 λ}, PlotRange → {-A, A}, PlotLabel → t,
  AxesLabel → {"t (s)", "Amp (m)"}, LabelStyle → Larger],
 {t, 0, 10 T, 0.001, AnimationRate → 0.01, AnimationRunning → False}]
```

Out[47]=



(c). Find the transverse displacement of a particle at x = 0.360, t = 0.150.

In[48]:=
```
xc = 0.360; tc = 0.150;
```

In[49]:=
```
y[xc, tc]
```

Out[49]= 0.0494975

In[50]:=
```
Show[{Plot[y[x, tc], {x, 0, 3 λ}, PlotRange → {-A, A}, PlotLabel → tc],
  Graphics[{Red, Disk[{xc, y[xc, tc]}, {0.02, 0.004}]}]},
 AxesLabel → {"x (m)", "Amp (m)"}, LabelStyle → Larger, ImageSize → Scaled[0.75]
]
```

Out[50]=



(d). Look at x = 0.360. When does it next reach the maximum?

In[51]:=
```
nf[x_] := PaddedForm[Chop[x], {4, 3}] (* Handy formatting function *)
```

In[52]:=
```
Manipulate[Show[{Plot[y[x, t], {x, 0, 3 λ},
    PlotRange → {-A, A}, PlotLabel → {nf[t], nf[xc], nf[y[xc, t]]}],
   Graphics[{Red, Disk[{xc, y[xc, t]}, {0.02, 0.004}]}]},
  AxesLabel → {"t (s)", "Amp (m)"}, LabelStyle → Larger, ImageSize → Scaled[0.75]
 ], {{t, 0.15}, 0, 10 T, 0.001, Appearance → "Open", AnimationRate → 0.01}]
```

Out[52]=



There are many times at which y[xc] will be the maximum. You need to tell *Mathematica* where to start looking. From the animation, it appears to be about 0.155 s.

In[53]:=
```
FindRoot[y[xc, t] == A, {t, tc}]
```

Out[53]= {t → 0.155}

The difference from part (c) is thus about 0.005 s.

In[54]:=
```
t - tc /. %
```

Out[54]= 0.00499999

The next time y[xc] is a maximum is at t = 0.195 s.

In[55]:=
```
Δt = 0.195 - 0.155
```

Out[55]= 0.04

That should be one period.

In[56]:=
```
T == Δt
```

Out[56]= True